
ECE 545—Digital System Design with VHDL
Lecture 5

Behavioral VHDL Coding (for Synthesis), 
Structural VHDL Coding

9/23/08

Outline

• Behavioral VHDL Coding for Synthesis
• Process Statement

• Sequential Logic: Latches, Flip-Flops, Registers, Shift 
Registers, Counters

• Behavioral VHDL Statements: If/Then, Wait, Case, For 
Loop, While Loop

• Testbenches

• Structural VHDL Coding
• Explicit component declaration versus packages

• Constants

Resources

• Volnei A. Pedroni, Circuit Design with VHDL

Chapter 6, Sequential Code

Chapter 10, Packages and Components

Chapter 7, Signals and Variables

Behavioral VHDL Coding for Synthesis

Combinational 

Logic

Combinational 

Logic

Registers

Register Transfer Level (RTL) Design Description

VHDL Design Styles (Architecture)

STRUCTURAL

components and
interconnects

VHDL Design 
Styles

DATAFLOW

“concurrent”
statements • Sequential Logic 

(registers,
counters, etc.)

• State machines 
•Complex Comb. 
Logic

“sequential” statements

NON-
SYNTHESIZABLE

SYNTHESIZABLE

BEHAVIORAL

• Test Benches
• Modeling IP• Gates

• Simple Comb. 
Logic

Behavioral Code

• Behavioral code also referred to as “sequential code”
• Behavioral code is characterized by PROCESS 

STATEMENTS
• In process statements, sections of code are executed 

sequentially—order matters!

• This is unlike dataflow (i.e. concurrent) code, in which 
order does not matter

8

Anatomy of a Process

[label:] process [(sensitivity list)]

[declaration part]

begin

statement part

end process [label];

OPTIONAL

Sensitivity List

• List of signals to which the process is 
sensitive.
• Whenever there is an event on any of the 

signals in the sensitivity list, the process 
fires.

• If an event occurs on a signal not listed 
in the sensitivity list, process does not 
fire

• Every time the process fires, it will run in its 
entirety.

• When designing synthesizable code, use a 
sensitivity list

• Process without a sensitivity list is used in a 
testbench
• Usually not synthesizable
• It will begin firing at the beginning of 

simulation
• When it finishes it will start at beginning of 

process (unless WAIT command issued)
• See Lecture 1 for more details

process(sensitivity list)
declaration part 

begin
statement part 

end process;

Declaration Part

• Declaration part (optional)
• Lists variables of the process; signals 

are declared in architecture declaration
• Variables versus signals

• Variables exist only locally in a process. 
To get variable’s value out of process, 
need to assign to signal

• Variable within a process is updated 
immediately; signal within a process is 
updated at the end of the process

• Variables use :=, signals use <=
• Read Pedroni Chapter 7 for more 

information
• Variables can be std_logic_vector, 

signed, unsigned, integer, etc.
• In this class only entity ports should be 

std_logic or std_logic_vector
• Variables, signals, generics can be 

other data types

process(sensitivity list)
declaration part 

begin
statement part 

end process;

Statement Part

• Statement Part
• Contains sequential statements to be 

executed each time the process Is 
activated

• Analogous to conventional 
programming languages process(sensitivity list)

declaration part 
begin

statement part 
end process;

WITHIN THE STATEMENT PART, THE 
SEQUENCE OF OPERATIONS MATTERS!!

Processes in VHDL

• Processes Describe Sequential Behavior
• Processes in VHDL Are Very Powerful Statements

• Allow to define an arbitrary behavior that may be difficult 
to represent by a real circuit

• Not every process can be synthesized

• Use Processes with Caution in the Code to Be 
Synthesized

• Use Processes Freely in Testbenches

Latches, Flip-Flops, Registers

14

D Latch vs. D Flip-Flop

Latch transparent when clock is high

“Samples” D on rising edge of clock

D Q

CLK

D Q

CLK

D

CLK

Q

D

CLK

Q

LIBRARY ieee ; 

USE ieee.std_logic_1164.all ; 

ENTITY latch IS 

PORT ( D, Clock : IN STD_LOGIC ; 

Q : OUT STD_LOGIC) ; 

END latch ;

ARCHITECTURE Behavior OF latch IS    

BEGIN

PROCESS ( D, Clock ) 

BEGIN

IF Clock = '1' THEN

Q <= D ; 

END IF ; 

END PROCESS ; 

END Behavior; 

D Q 

Clock 

Process is sensitive to D and clock

D Latch

LIBRARY ieee ; 

USE ieee.std_logic_1164.all ; 

ENTITY flipflop IS 

PORT ( D, Clock : IN STD_LOGIC ; 

Q : OUT STD_LOGIC) ; 

END flipflop ;

ARCHITECTURE Behavior_1 OF flipflop IS    

BEGIN

PROCESS ( Clock ) 

BEGIN 

IF ( Clock'EVENT AND Clock = '1') THEN 

Q <= D ; 

END IF ; 

END PROCESS ; 

END Behavior_1 ; 

D Q 

Clock 

Process sensitive to clock only
(changes in D do not matter is clock is not changing)

D Flip-Flop: Standard Coding Style

LIBRARY ieee ; 

USE ieee.std_logic_1164.all ; 

ENTITY flipflop IS 

PORT ( D, Clock : IN STD_LOGIC ; 

Q : OUT STD_LOGIC) ; 

END flipflop ;

ARCHITECTURE Behavior_1 OF flipflop IS    

BEGIN

PROCESS ( Clock ) 

BEGIN 

IF rising_edge(Clock) THEN 

Q <= D ; 

END IF ; 

END PROCESS ; 

END Behavior_1 ; 

D Q 

Clock 

D Flip-Flop: Alternate Coding Style #1

LIBRARY ieee ; 

USE ieee.std_logic_1164.all ; 

ENTITY flipflop IS 

PORT ( D, Clock : IN STD_LOGIC ; 

Q : OUT STD_LOGIC) ; 

END flipflop ;

ARCHITECTURE Behavior_2 OF flipflop IS    

BEGIN

PROCESS

BEGIN 

WAIT UNTIL Clock'EVENT AND Clock = '1' ;

Q <= D ; 

END PROCESS ; 

END Behavior_2 ; 

D Q 

Clock 

D Flip-Flop: Alternate Coding Style #2

LIBRARY ieee ; 

USE ieee.std_logic_1164.all ; 

ENTITY flipflop IS 

PORT ( D, Clock : IN STD_LOGIC ; 

Q : OUT STD_LOGIC) ; 

END flipflop ;

ARCHITECTURE Behavior_2 OF flipflop IS    

BEGIN

PROCESS

BEGIN 

WAIT UNTIL rising_edge(Clock) ;

Q <= D ; 

END PROCESS ; 

END Behavior_2 ; 

D Q 

Clock 

D Flip-Flop: Alternate Coding Style #3

LIBRARY ieee ; 

USE ieee.std_logic_1164.all ; 

ENTITY flipflop IS 

PORT ( D, Resetn , Clock : IN STD_LOGIC ; 

Q : OUT STD_LOGIC) ; 

END flipflop ;

ARCHITECTURE Behavior OF flipflop IS    

BEGIN

PROCESS ( Resetn , Clock ) 

BEGIN 

IF Resetn = '0' THEN 

Q <= '0' ; 

ELSIF (Clock'EVENT AND Clock = '1') THEN 

Q <= D ; 

END IF ; 

END PROCESS ;

END Behavior ; 

D Q 

Clock 

Resetn

D flip-flop with asynchronousreset

Bubble means 
active-low reset

(written resetn for
reset “not”)

LIBRARY ieee ; 

USE ieee.std_logic_1164.all ; 

ENTITY flipflop IS 

PORT ( D, Resetn , Clock : IN STD_LOGIC ; 

Q : OUT STD_LOGIC) ; 

END flipflop ;

ARCHITECTURE Behavior OF flipflop IS    

BEGIN

PROCESS (Clock )

BEGIN 

IF (Clock'EVENT AND Clock = '1') THEN 

IF (Resetn = '0') THEN 

Q <= '0' ; 

ELSE

Q <= D ; 

END IF ; 

END IF ;

END PROCESS ;

END Behavior ; 

D Q 

Clock 

Resetn

D flip-flop with synchronousreset

LIBRARY ieee ;

USE ieee.std_logic_1164.all ;

ENTITY reg8 IS

PORT ( D : IN STD_LOGIC_VECTOR(7 DOWNTO 0) ;

Resetn, Clock : IN STD_LOGIC ;

Q : OUT STD_LOGIC_VECTOR(7 DOWNTO 0) ) ;

END reg8 ;

ARCHITECTURE Behavior OF reg8 IS

BEGIN

PROCESS ( Resetn, Clock )

BEGIN

IF (Resetn = '0') THEN

Q <= "00000000" ;

ELSIF (Clock'EVENT AND Clock = '1') THEN

Q <= D ;

END IF ;

END PROCESS ;

END Behavior ;

Resetn

Clock

reg8

8 8

D Q

8-bit register with asynchronous reset 

Recall: Register is just a collection of D flip-flops

LIBRARY ieee ;

USE ieee.std_logic_1164.all ;

ENTITY regn IS

GENERIC ( N : INTEGER := 16 ) ;

PORT ( D : IN STD_LOGIC_VECTOR( N-1 DOWNTO 0) ;

Resetn, Clock : IN STD_LOGIC ;

Q : OUT STD_LOGIC_VECTOR(N-1 DOWNTO 0) ) ;

END regn ;

ARCHITECTURE Behavior OF regn IS

BEGIN

PROCESS ( Resetn, Clock )

BEGIN

IF (Resetn = '0') THEN

Q <= (OTHERS => '0') ;

ELSIF (Clock'EVENT AND Clock = '1') THEN

Q <= D ;

END IF ;

END PROCESS ;

END Behavior ;

Resetn

Clock

regn

N N

D Q

N-bit register with asynchronous reset

A word on generics

• Generics are typically integervalues
• In this class the entity inputs and outputs should be std_logic or 

std_logic_vector
• But the generics can beinteger

• Generics are given a default value
• GENERIC ( N : INTEGER := 16 );
• This value can be overwritten when entity is instantiated as a 

component

• Generics are very useful when instantiating an often-used 
component
• Need a 32-bit register in one place, and 16-bit register in another
• Can use the same generic code, just configure them differently

LIBRARY ieee ;

USE ieee.std_logic_1164.all ;

ENTITY regn IS

GENERIC ( N : INTEGER := 8 ) ;

PORT ( D : IN STD_LOGIC_VECTOR(N-1 DOWNTO 0) ;

Enable , Clock : IN STD_LOGIC ;

Q : OUT STD_LOGIC_VECTOR(N-1 DOWNTO 0) ) ;

END regn ;

ARCHITECTURE Behavior OF regn IS

BEGIN

PROCESS (Clock)

BEGIN

IF (Clock'EVENT AND Clock = '1' ) THEN

IF (Enable = '1') THEN

Q <= D ;

END IF ;

END IF;

END PROCESS ;

END Behavior ;

QD

Enable

Clock

regn

N N

N-bit register with synchronous enable

Asychronous vs. Synchronous

• In the IF loop, asynchronous items are
• Before the Clock'EVENT AND Clock = ‘1’statement

• In the IF loop, synchronous items are
• After the Clock'EVENT AND Clock = ‘1’statement

Counters

LIBRARY ieee ;

USE ieee.std_logic_1164.all ;

USE ieee.std_logic_unsigned.all ; --can also declare ieee.std_logic_arith.all but not needed

ENTITY upcount IS

PORT ( Clear , Clock : IN STD_LOGIC ;

Q : BUFFER STD_LOGIC_VECTOR(1 DOWNTO 0) ) ;

END upcount ;

ARCHITECTURE Behavior OF upcount IS

BEGIN

upcount: PROCESS ( Clock )

BEGIN

IF (Clock'EVENT AND Clock = '1') THEN

IF (Clear = '1') THEN

Q <= "00" ;

ELSE

Q <= Q + "01" ;

END IF ;

END IF;

END PROCESS;

END Behavior ;

Q
Clear

Clock

upcount

2

DON’T USE BUFFER

2-bit up-counter with synchronousreset

LIBRARY ieee ;

USE ieee.std_logic_1164.all ;

USE ieee.std_logic_unsigned.all ; --can also declare ieee.std_logic_arith.all but not needed

ENTITY upcount IS

PORT ( Clear , Clock : IN STD_LOGIC ;

Q : OUT STD_LOGIC_VECTOR(1 DOWNTO 0) ) ;

END upcount ;

ARCHITECTURE Behavior OF upcount IS

SIGNAL Count : std_logic_vector(1 downto 0);

BEGIN

upcount: PROCESS ( Clock )

BEGIN

IF (Clock'EVENT AND Clock = '1') THEN

IF (Clear = '1') THEN

Count <= "00" ;

ELSE

Count <= Count + 1 ;

END IF ;

END IF;

END PROCESS;

Q <= Count;

END Behavior ;

Q
Clear

Clock

upcount

2

Assigned outside the process

2-bit up-counter with synchronousreset

LIBRARY ieee ;

USE ieee.std_logic_1164.all ;

USE ieee.std_logic_unsigned.all ;

ENTITY upcount IS

PORT ( Clock, Resetn , Enable : IN STD_LOGIC ;

Q : OUT STD_LOGIC_VECTOR (3 DOWNTO 0)) ;

END upcount ;

Q

Enable

Clock
upcount

4

Resetn

4-bit up-counter with asynchronousreset (1)

ARCHITECTURE Behavior OF upcount IS

SIGNAL Count : STD_LOGIC_VECTOR (3 DOWNTO 0) ;

BEGIN

PROCESS ( Clock, Resetn )

BEGIN

IF Resetn = '0' THEN

Count <= "0000" ;

ELSIF (Clock'EVENT AND Clock = '1') THEN

IF (Enable = '1') THEN

Count <= Count + 1 ;

END IF ;

END IF ;

END PROCESS ;

Q <= Count ;

END Behavior ;

Q

Enable

Clock
upcount

4

Resetn

4-bit up-counter with asynchronousreset (2)

Shift Registers

Shift register

D Q
Sin

Clock

D Q D Q D Q

Q(3) Q(2) Q(1) Q(0)

Enable

34

Shift Register With Parallel Load

D(3)

D Q

Clock

Enable

Sin
D(2)

D Q

D(1)

D Q

D(0)

D Q

Q(0)Q(1)Q(2)Q(3)

Load

LIBRARY ieee ;

USE ieee.std_logic_1164.all ;

ENTITY shift4 IS

PORT ( D : IN STD_LOGIC_VECTOR(3 DOWNTO 0) ;

Enable : IN STD_LOGIC ;

Load : IN STD_LOGIC ;

Sin : IN STD_LOGIC ;

Clock : IN STD_LOGIC ;

Q : OUT STD_LOGIC_VECTOR(3 DOWNTO 0) ) ;

END shift4 ;

Q

Enable

Clock
shift4

4

D

Load

Sin

4

4-bit shift register with parallel load (1)

ARCHITECTURE Behavior_1 OF shift4 IS

SIGNAL Qt : STD_LOGIC_VECTOR(3 DOWNTO 0); 

BEGIN

PROCESS (Clock)

BEGIN

IF (Clock'EVENT AND Clock = '1')  THEN

IF (Load = '1') THEN

Qt <= D ;

ELSIF (Enable = '1') THEN

Qt(0) <= Qt(1) ;

Qt(1) <= Qt(2); 

Qt(2) <= Qt(3) ; 

Qt(3) <= Sin; 

END IF ;

END IF ;

END PROCESS ;

Q <= Qt;

END Behavior_1 ;

Q

Enable

Clock
shift4

4

D

Load

Sin

4

Usually
Load is before
Enable

4-bit shift register with parallel load (2)

LIBRARY ieee ;

USE ieee.std_logic_1164.all ;

ENTITY shiftn IS

GENERIC ( N : INTEGER := 8 ) ;

PORT ( D : IN STD_LOGIC_VECTOR( N-1 DOWNTO 0) ;

Enable : IN STD_LOGIC ;

Load : IN STD_LOGIC ;

Sin : IN STD_LOGIC ;

Clock : IN STD_LOGIC ;

Q : OUT STD_LOGIC_VECTOR(N-1 DOWNTO 0) ) ;

END shiftn ;

Q

Enable

Clock
shiftn

N

D

Load

Sin

N

N-bit shift register with parallel load (1)

ARCHITECTURE Behavior OF shiftn IS

SIGNAL Qt: STD_LOGIC_VECTOR( N-1 DOWNTO 0);

BEGIN

PROCESS (Clock)

BEGIN

IF (Clock'EVENT AND Clock = '1' ) THEN

IF (Load = '1') THEN

Qt <= D ;

ELSIF (Enable = '1') THEN

Genbits: FOR i IN 0 TO N-2 LOOP

Qt(i) <= Qt(i+1) ;

END LOOP ;

Qt(N-1) <= Sin ;

END IF;

END IF ;

END PROCESS ;

Q <= Qt;

END Behavior ;

Q

Enable

Clock
shiftn

N

D

Load

Sin

N

N-bit shift register with parallel load (2)

ASMs and synchronous signals

• For ASM counters and shift registers, helpful to use 
synchronous loads and enables
• Easier to understanding timing

40

Behavioral VHDL Statements

If Statement

• If statement

• elseand elsifare optional
• Most commonly used type of statement

if boolean expression then
statements

elsifboolean expressionthen
statements

else boolean expressionthen
statements

end if;

SELECTOR: process

begin

WAIT UNTIL Clock'EVENT AND Clock = '1' ;

IF Sel = "00" THEN

f  <= x1;

ELSIF Sel = "10" THEN 

f  <= x2;

ELSE

f  <= x3;

END IF;

end process;

If Statement-Example

• WAIT UNTIL
• Syntax: WAIT UNTIL signal_condition;

• Explanation: Accepts one signal, waits until signal condition before going to 
next statement in the process.

• Example: WAIT UNTIL (clk'EVENT and clk='1');

• Synthesizable: Yes (Can use in synthesizable code and testbenches)
• WAIT ON

• Syntax: WAIT ON signal1 [,signal2,…];

• Explanation: Accepts multiple signals, waits until any of the signals change.
• Example: WAIT ON clk,rst;

• Synthesizable: Yes (Can use in synthesizable code and testbenches)
• WAIT FOR

• Syntax: WAIT FOR time 

• Explanation: Wait for a certain amount of time before going to next statement 
in process.

• Synthesizable: No (Can use in testbenches only)
• WAIT (i.e. wait forever)

• Syntax: WAIT

• Explanation: Waits forever, i.e. never goes to the next statement in process.
• Synthesizable: No (Can use in testbenches only)

Wait Statements

TESTBENCHES

ONLY

Case Statement

• Case statement
• Choices have to cover all

possible values of the 
condition
• Use othersto specify all 

remaining cases

casecondition is
when choice_1=>

statements
when choice_2=>

statements

when others =>
statements

end case;

• For Loop Statement

• Repeats a section of VHDL code sequentially (i.e. 
order matters)

• This is different from generate statement which is 
concurrent (order does not matter)

For Loop

[label:] fori inrange loop
statements

end loop [label];

While Loop

• While Loop Statement

• Repeats a section of VHDL code sequentially (i.e. 
order matters)

• Checks a condition before execution

[label:] while conditionloop
statements

end loop [label];

Sequential Logic For Beginners

• Use processes with very simple structure only to describe
• Registers
• Shift registers
• Counters
• State machines

• Use examples discussed in class as a template.

• Create generic entities for registers, shift registers, and 
counters, and instantiate the corresponding components in a 
higher level circuit using GENERIC MAP PORT MAP.

• Supplement sequential components with combinational logic 
described using concurrent statements

Sequential Logic For Intermediates (1)

1. Use Processes with IF and CASE statements only. Do not use 
LOOPS or VARIABLES.

2. Sensitivity list of the PROCESS should include onlysignals that can 
by themselves change the outputs of the sequential circuit (typically, 
clock and asynchronous set or reset)

3. Do not use PROCESSes without sensitivity list
(they can be synthesizable, but make simulation inefficient)

Sequential Logic For Intermediates (2)

Given a single signal, the assignments to this signal should 
only be made within a single process block in order to avoid
possible conflicts in assigning values to this signal.

Process 1: PROCESS (a, b)
BEGIN

y <= a AND b;
END PROCESS;

Process 2: PROCESS (a, b)
BEGIN

y <= a OR b;
END PROCESS;

STATEMENTS WITHIN A 
PROCESS ARE SEQUENTIAL...

BUT TWO PROCESSES ARE
CONCURRENT TO ONE ANOTHER.

THEY RUN IN PARALLEL

Sequential Logic For Advanced

Describe the algorithm you are trying to
implement in pseudocode.

Translate the pseudocode directly to VHDL
using processes with IF, CASE, LOOP, 
and VARIABLES.

Multiple precautions needed, 
Template and examples covered later 
in class.

Testbenches

Review: Generating selected values of one input

SIGNAL test_vector : STD_LOGIC_VECTOR(2 downto 0);

BEGIN

.......

testing: PROCESS 

BEGIN

test_vector <= "000";

WAIT FOR 10 ns;

test_vector <= "001";

WAIT FOR 10 ns;

test_vector <= "010";

WAIT FOR 10 ns;

test_vector <= "011";

WAIT FOR 10 ns;

test_vector <= "100";

WAIT FOR 10 ns;

END PROCESS;

........

END behavioral;

SIGNAL test_vector : STD_LOGIC_VECTOR(3 downto 0) :="0000";

BEGIN

.......

testing: PROCESS

BEGIN

WAIT FOR 10 ns;

test_vector <= test_vector + 1; 

end process TESTING;

........

END behavioral;

MUST HAVE
INITIAL VALUE

Review: Generating all values of one input

SIGNAL test_ab : STD_LOGIC_VECTOR(1 downto 0);

SIGNAL test_sel : STD_LOGIC_VECTOR(1 downto 0);

BEGIN

.......

double_loop: PROCESS
BEGIN

test_ab  <="00";
test_sel  <="00";
for I in 0 to 3 loop

for J in 0 to 3 loop
wait for 10 ns;

test_ab <= test_ab + 1;
end loop;
test_sel <= test_sel + 1;

end loop;
END PROCESS;

........

END behavioral;

CAN PUT INITIAL VALUE
IN SIGNAL DECLARATION
ORIN PROCESS ITSELF

Review: Generating all values of one input

Simple Testbenches for Sequential Logic

• Use a process without sensitivity list to perform main 
sequential simulation
• reset assert/deassert

• change data on input ports

• Use a concurrent statement to create a clock
• This will run in parallel with the process statement

• Later we may use multiple processes 

SIGNAL clk : std_logic := '0'; -- initial value

CONSTANT clk_period : TIME := 20 ns; -- 50 MHz

BEGIN

.......

clk <= not clk after (clk_period/2);

.......

END behavioral;

Generating a clock

Input data in relation to clock

CLK

• Flip-flops (and in turn registers, shift registers, and counters) have a 
setup time
• Data must be stable before setup time or else may get an error loading data
• This will make a big difference when we do post-place-and-route timing 

simulations
• This is also true for synchronous enables and loads

• In test benches, if input data changes on the negative edgeof the 
clock, you should normally not violate setup time of the next 
positive edge
• If violate, then can use a delay from the positive edge (more on this later)

setup timeInput data to 
flip-flop 
must be 
stable 
before here

setup time

use negative edge of clock

Testbench: N-bit register with asynchronous reset

library ieee;

use ieee.std_logic_1164.all;   

use ieee.std_logic_unsigned.all; -- using addition in for loop

entity regn_tb is

generic(

N : INTEGER := 4 ); -- testing a 4-bit register, not 16-bit

end regn_tb;

architecture TB_ARCHITECTURE of regn_tb is

component regn

generic(

N : INTEGER := 16 );

port(

D : in std_logic_vector(N-1 downto 0);

Resetn : in std_logic;

Clock : in std_logic;

Q : out std_logic_vector(N-1 downto 0) );

end component;

Testbench cont'd

signal D : std_logic_vector(N-1 downto 0);

signal Resetn : std_logic;

signal Clock : std_logic := '0'; -- initialize clock

constant clockperiod : time := 20 ns;

signal Q : std_logic_vector(N-1 downto 0);

begin

UUT : regn

generic map (

N => N -- 4-bit register

)

port map (

D => D,

Resetn => Resetn,

Clock => Clock,

Q => Q

);

60

Testbench cont'd
process

begin

resetn <= '0'; -- put system in reset

wait until (Clock'event and Clock='0'); -- negedge of clock

wait for (clockperiod*2); -- wait two more clock cycles 

resetn <= '1'; -- deassert reset

D <= "0000"; -- input changes on negedge of clock

for i in 0 to 15 loop  

wait for clockperiod;

D <= D + 1;

end loop;

wait; -- wait forever

end process;

Clock <= NOT Clock after (clockperiod/2);

end TB_ARCHITECTURE;

Waveform

D changes on negative edge so no
violation of setup time

Processes in test benches

• Two processes will run concurrently
• Processes are sequential within a process

• Processes themselves are concurrent

BEGIN

.......

generator1: PROCESS

BEGIN

reset <= '1';

WAIT FOR 100 ns;

reset <= '0';

test_vector <="000";

WAIT;

END PROCESS;

generator2: PROCESS

BEGIN

WAIT FOR 200 ns;

test_vector <="001";

WAIT FOR 600 ns;

test_vector <="011";

END PROCESS;

.......

END behavioral;

Typical error

Non-synthesizable VHDL
Used in Test-Benches

Delays

Delays are not synthesizable

Statements, such as

wait for 5 ns

a <= b after10 ns

will not produce the required delay, and 

should not be used in the code intended

for synthesis.

Initializations

Declarations of signals (and variables)

with initialized values, such as

SIGNAL  a : STD_LOGIC := ‘0’;

cannot be synthesized, and thus should

be avoided.

If present, they will be ignored by the

synthesis tools.

Use set and reset signals instead.

Structural VHDL Coding

VHDL Design Styles (Architecture)

STRUCTURAL

components and
interconnects

VHDL Design 
Styles

DATAFLOW

“concurrent”
statements • Sequential Logic 

(registers,
counters, etc.)

• State machines 
•Complex Comb. 
Logic

“sequential” statements

NON-
SYNTHESIZABLE

SYNTHESIZABLE

BEHAVIORAL

• Test Benches
• Modeling IP• Gates

• Simple Comb. 
Logic

• component instantiation      (port map)

• generate scheme for component instantiations

(for-generate)

• component instantiation with generic 

(generic map, port map)

Major instructions

Structural VHDL

• component instantiation      (port map)

• component instantiation with generic 

(generic map, port map)

• generate scheme for component instantiations

(for-generate)

Major instructions

Structural VHDL

Components

• A component is an instantiation of an entity(plus 
an architecture)

• Allows a designer to re-use common pieces of code 
(i.e. registers, counters, etc.)

• Two ways of declaring components
• METHOD #1: Explicit component declaration

• Components declared in main code

• METHOD #2: Package component declaration
• Components declared in a package

• Component declaration tells the compiler the 
ports of the components about to be instantiated

Circuit built of medium scale components

2-to-1 Multiplexer

(a) Graphical symbol (b) Truth table

VHDL code for a 2-to-1 Multiplexer

LIBRARY ieee ;

USE ieee.std_logic_1164.all ;

ENTITY mux2to1 IS

PORT ( w0, w1, s : IN STD_LOGIC ;

f : OUT STD_LOGIC ) ;

END mux2to1 ;

ARCHITECTURE dataflow OF mux2to1 IS

BEGIN

f <= w0 WHEN s = '0' ELSE w1 ;

END dataflow ;

Priority Encoder

VHDL code for a Priority Encoder
LIBRARY ieee ;

USE ieee.std_logic_1164.all ;

ENTITY priority IS

PORT ( w : IN STD_LOGIC_VECTOR(3 DOWNTO 0) ;

y : OUT STD_LOGIC_VECTOR(1 DOWNTO 0) ;

z : OUT STD_LOGIC ) ;

END priority ;

ARCHITECTURE dataflow OF priority IS

BEGIN

y <= "11" WHEN w(3) = '1' ELSE 

"10" WHEN w(2) = '1' ELSE

"01" WHEN w(1) = '1' ELSE

"00" ;

z <= '0' WHEN w = "0000" ELSE '1' ;

END dataflow ;

2-to-4 Decoder

VHDL code for a 2-to-4 Decoder
LIBRARY ieee ;

USE ieee.std_logic_1164.all ;

ENTITY dec2to4 IS

PORT ( w : IN STD_LOGIC_VECTOR(1 DOWNTO 0) ;

En : IN STD_LOGIC ;

y : OUT STD_LOGIC_VECTOR(3 DOWNTO 0) ) ;

END dec2to4 ;

ARCHITECTURE dataflow OF dec2to4 IS

SIGNAL Enw : STD_LOGIC_VECTOR(2 DOWNTO 0) ;

BEGIN

Enw <= En & w ;

WITH Enw SELECT

y <= "0001" WHEN "100",

"0010" WHEN "101",

"0100" WHEN "110",

"1000" WHEN "111",

"0000" WHEN OTHERS ;

END dataflow ;

LIBRARY ieee ;

USE ieee.std_logic_1164.all ;

ENTITY regn IS

GENERIC ( N : INTEGER := 8 ) ;

PORT ( D : IN STD_LOGIC_VECTOR(N-1 DOWNTO 0) ;

Enable, Clock : IN STD_LOGIC ;

Q : OUT STD_LOGIC_VECTOR(N-1 DOWNTO 0) ) ;

END regn ;

ARCHITECTURE Behavior OF regn IS

BEGIN

PROCESS (Clock)

BEGIN

IF (Clock'EVENT AND Clock = '1' ) THEN

IF Enable = '1' THEN

Q <= D ;

END IF ;

END IF;

END PROCESS ;

END Behavior ;

QD

Enable

Clock

regn

N N

N-bit register with enable

Circuit built of medium scale components

METHOD #1: Explicit component declaration

• Components declared in main code
• Actual instantiations and port maps always in main 

code

82

Structural description – example (1)

LIBRARY ieee ;
USE ieee.std_logic_1164.all ;

ENTITY priority_resolver IS
PORT (r : IN STD_LOGIC_VECTOR(5 DOWNTO 0) ;

s : IN STD_LOGIC_VECTOR(1 DOWNTO 0) ;
clk : IN      STD_LOGIC;
en        : IN      STD_LOGIC;
t : OUT STD_LOGIC_VECTOR(3 DOWNTO 0) ) ;

END priority_resolver;

ARCHITECTURE structural OF priority_resolver IS

SIGNAL  p : STD_LOGIC_VECTOR (3 DOWNTO 0) ;
SIGNAL  q : STD_LOGIC_VECTOR (1 DOWNTO 0) ;
SIGNAL  z : STD_LOGIC_VECTOR (3 DOWNTO 0) ;
SIGNAL  ena : STD_LOGIC ;

Structural description – example (2)
COMPONENT mux2to1

PORT (w0, w1, s : IN STD_LOGIC ;
f : OUT STD_LOGIC ) ;

END COMPONENT ;

COMPONENT priority
PORT (w : IN STD_LOGIC_VECTOR(3 DOWNTO 0) ;

y : OUT STD_LOGIC_VECTOR(1 DOWNTO 0) ;
z : OUT STD_LOGIC ) ;

END COMPONENT ;

COMPONENT dec2to4
PORT (w : IN STD_LOGIC_VECTOR(1 DOWNTO 0) ;

En : IN STD_LOGIC ;
y : OUT STD_LOGIC_VECTOR(3 DOWNTO 0) ) ;

END COMPONENT ;

COMPONENT regn
GENERIC ( N : INTEGER := 8 ) ; -- default size
PORT ( D     : IN STD_LOGIC_VECTOR(N-1 DOWNTO 0) ;

Enable, Clock : IN STD_LOGIC ;
Q     : OUT STD_LOGIC_VECTOR(N-1 DOWNTO 0) ) ;

END COMPONENT ;

84

BEGIN

u1: mux2to1 PORT MAP ( w0 => r(0) ,
w1 => r(1),
s => s(0),
f => p(0));

p(1) <= r(2);
p(2) <= r(3);

u2: mux2to1 PORT MAP ( w0 => r(4) ,
w1 => r(5),
s => s(1),
f => p(3));

u3: priority PORT MAP ( w => p,
y => q,
z => ena);

u4: dec2to4 PORT MAP ( w => q,
En => ena,
y => z);

u5: regn GENERIC MAP ( N => 4 )

PORT MAP ( D => z ,

Enable => En ,

Clock => Clk,

Q => t );

END structural;

Structural description – example (3)

Entity Port Name

Local Wire/Port

Change size of regn
from default size (8) to
smaller size (4)

Named association connectivity

• recommended in majority of cases,
prevents ommisions and mistakes

COMPONENT dec2to4
PORT (w : IN STD_LOGIC_VECTOR(1 DOWNTO 0) ;

En : IN STD_LOGIC ;
y : OUT STD_LOGIC_VECTOR(0 TO 3) ) ;

END COMPONENT ;

u4: dec2to4 PORT MAP (w   => q,
En => ena,
y => z);

Positional association connectivity

• Allowed for cases of  
• small number of ports 

• multiple instantiations of the same component, in regular 
structures

COMPONENT dec2to4
PORT (w : IN STD_LOGIC_VECTOR(1 DOWNTO 0) ;

En : IN STD_LOGIC ;
y : OUT STD_LOGIC_VECTOR(0 TO 3) ) ;

END COMPONENT ;

u4: dec2to4 PORT MAP (q, ena, z);

Structural description with
positional association connectivity

BEGIN

u1: mux2to1 PORT MAP (r(0), r(1), s(0), p(0));

p(1) <= r(2);

p(1) <= r(3);

u2: mux2to1 PORT MAP (r(4) , r(5), s(1), p(3));

u3: priority PORT MAP (p, q, ena);

u4: dec2to4 PORT MAP (q, ena, z);

u5: regn GENERIC MAP(4) PORT MAP (z, En, Clk, t);

END structural;

• For simple projects put entity .vhd
files all in same directory

• Declare components in main code
• If using Aldec, make sure compiler 

knows the correct hierarchy
• From lowest to highest

• Xilinx will figure out hierarchy 
automatically

Explicit Component Declaration Tips

METHOD #2: Package component declaration

• Components declared in package
• Actual instantiations and port maps always in main 

code

Packages

• Instead of declaring all components can declare all 
components in a PACKAGE, and INCLUDE the 
package once
• This makes the top-level entity code cleaner

• It also allows that complete package to be used by 
another designer

• A package can contain
• Components

• Functions, Procedures

• Types, Constants

Package – example (1)

LIBRARY ieee ;
USE ieee.std_logic_1164.all ;

PACKAGE GatesPkg IS

COMPONENT mux2to1
PORT (w0, w1, s : IN STD_LOGIC ;

f : OUT STD_LOGIC ) ;
END COMPONENT ;

COMPONENT priority
PORT (w : IN STD_LOGIC_VECTOR(3 DOWNTO 0) ;

y : OUT STD_LOGIC_VECTOR(1 DOWNTO 0) ;
z : OUT STD_LOGIC ) ;

END COMPONENT ;

Package – example (2)

COMPONENT dec2to4
PORT (w : IN STD_LOGIC_VECTOR(1 DOWNTO 0) ;

En : IN STD_LOGIC ;
y : OUT STD_LOGIC_VECTOR(0 TO 3) ) ;

END COMPONENT ;

COMPONENT regn
GENERIC ( N : INTEGER := 8 ) ;
PORT ( D     : IN STD_LOGIC_VECTOR(N-1 DOWNTO 0) ;

Enable, Clock : IN STD_LOGIC ;
Q     : OUT STD_LOGIC_VECTOR(N-1 DOWNTO 0) ) ;

END COMPONENT ;

constant ADDAB : std_logic_vector(3 downto 0) := "0 000";

constant ADDAM : std_logic_vector(3 downto 0) := "0 001";

constant SUBAB : std_logic_vector(3 downto 0) := "0 010";

constant SUBAM : std_logic_vector(3 downto 0) := "0 011";

constant NOTA : std_logic_vector(3 downto 0) := "01 00";

constant NOTB : std_logic_vector(3 downto 0) := "01 01";

constant NOTM : std_logic_vector(3 downto 0) := "01 10";

constant ANDAB : std_logic_vector(3 downto 0) := "0 111";

END GatesPkg;

Package – example (3)

Package usage (1)

LIBRARY ieee ;
USE ieee.std_logic_1164.all ;

USE work.GatesPkg.all;

ENTITY priority_resolver1 IS
PORT (r : IN STD_LOGIC_VECTOR(5 DOWNTO 0) ;

s : IN STD_LOGIC_VECTOR(1 DOWNTO 0) ;
clk : IN         STD_LOGIC;
en         : IN        STD_LOGIC;

t : OUT STD_LOGIC_VECTOR(3 DOWNTO 0) ) ;
END priority_resolver1;

ARCHITECTURE structural OF priority_resolver1 IS

SIGNAL  p : STD_LOGIC_VECTOR (3 DOWNTO 0) ;
SIGNAL  q : STD_LOGIC_VECTOR (1  DOWNTO 0) ;
SIGNAL  z : STD_LOGIC_VECTOR (3 DOWNTO 0) ;
SIGNAL  ena : STD_LOGIC ;

BEGIN

u1: mux2to1 PORT MAP ( w0 => r(0) ,
w1 => r(1),
s => s(0),
f => p(0));

p(1) <= r(2);
p(2) <= r(3);

u2: mux2to1 PORT MAP ( w0 => r(4) ,
w1 => r(5),
s => s(1),
f => p(3));

u3: priority PORT MAP ( w => p,
y => q,
z => ena);

u4: dec2to4 PORT MAP ( w => q,
En => ena,
y => z);

u5: regn GENERIC MAP ( N => 4 )

PORT MAP ( D => z ,

Enable => En ,

Clock => Clk,

Q => t );

END structural;

Package usage (2)

Aldec Compilation Order

• Include package 
before top-level

Explicit Component Declaration versus Package

• Explicit component declaration is when you declare 
components in main code
• When have only a few component declarations, this is 

fine
• When have many component declarations, use packages 

for readability

• Packages also help with portability and sharing of 
libraries among many users in a company

• Remember, the actual instantiations always take 
place in main code
• Only the declarations can be in main code or package

• component instantiation      (port map)
• component instantiation with generic 

(generic map, port map)
• generate scheme for component instantiations

(for-generate)

Major instructions

Structural VHDL

A 4-to-1 Multiplexer

LIBRARY ieee ;
USE ieee.std_logic_1164.all ;

ENTITY mux4to1 IS
PORT ( w0, w1, w2, w3 : IN STD_LOGIC ;

s : IN STD_LOGIC_VECTOR(1 DOWNTO 0) ;
f : OUT STD_LOGIC ) ;

END mux4to1 ;

ARCHITECTURE Dataflow OF mux4to1 IS
BEGIN

WITH s SELECT
f <= w0 WHEN "00",

w1 WHEN "01",
w2 WHEN "10",
w3 WHEN OTHERS ;

END Dataflow ;

Straightforward code for Example 1

LIBRARY ieee ;
USE ieee.std_logic_1164.all ;

ENTITY Example1 IS
PORT (  w : IN STD_LOGIC_VECTOR(0 TO 15) ;

s : IN STD_LOGIC_VECTOR(3 DOWNTO 0) ;
f : OUT STD_LOGIC ) ;

END Example1 ;
ARCHITECTURE Structure OF Example1 IS

COMPONENT mux4to1
PORT ( w0, w1, w2, w3 : IN STD_LOGIC ;

s : IN STD_LOGIC_VECTOR(1 DOWNTO 0) ;
f : OUT STD_LOGIC ) ;

END COMPONENT ;

SIGNAL m : STD_LOGIC_VECTOR(0 TO 3) ;

BEGIN
Mux1: mux4to1 PORT MAP ( w(0),    w(1),    w(2),    w(3),       s(1 DOWNTO 0), m(0) ) ;
Mux2: mux4to1 PORT MAP ( w(4),    w(5),    w(6),    w(7),       s(1 DOWNTO 0), m(1) ) ;
Mux3: mux4to1 PORT MAP ( w(8),    w(9),    w(10),  w(11),     s(1 DOWNTO 0), m(2) ) ;
Mux4: mux4to1 PORT MAP ( w(12),  w(13),  w(14),  w(15),     s(1 DOWNTO 0), m(3) ) ;
Mux5: mux4to1 PORT MAP ( m(0),    m(1),    m(2),   m(3),       s(3 DOWNTO 2),   f  ) ;

END Structure ;

Modified code for Example 1

ARCHITECTURE Structure OF Example1 IS

COMPONENT mux4to1

PORT ( w0, w1, w2, w3 : IN STD_LOGIC ;

s : IN STD_LOGIC_VECTOR(1 DOWNTO 0) ;

f : OUT STD_LOGIC ) ;

END COMPONENT ;

SIGNAL m : STD_LOGIC_VECTOR(0 TO 3) ;

BEGIN

G1: FOR i IN 0 TO 3 GENERATE

Muxes: mux4to1 PORT MAP (

w(4*i), w(4*i+1), w(4*i+2), w(4*i+3), s(1 DOWNTO 0) , m(i) ) ;

END GENERATE ;

Mux5: mux4to1 PORT MAP ( m(0), m(1), m(2), m(3), s( 3 DOWNTO 2), f ) ;

END Structure ;

Constants

Constants

Syntax:

CONSTANT   name : type := value;

Examples:

CONSTANT init_value : STD_LOGIC_VECTOR(3 downto 0) := "0100";
CONSTANT ANDA_EXT : STD_LOGIC_VECTOR(7 downto 0) :=  X"B4";
CONSTANT counter_width : INTEGER := 16;
CONSTANT buffer_address : INTEGER := 16#FFFE#;
CONSTANT clk_period : TIME := 20 ns;
CONSTANT strobe_period : TIME := 333.333 ms;

Constants - Features

Constants can be declared in a
PACKAGE, ENTITY, ARCHITECTURE

When declared in a PACKAGE, the constant
is truly global, for the package can be used
in several entities.

When declared in an ARCHITECTURE, the
constant is local, i.e., it is visible only within this architecture.

When declared in an ENTITY declaration, the constant 
can be used in all architectures associated with this entity.

106

Mixing Design Styles in an Architecture

VHDL Design Styles (Architecture)

STRUCTURAL

components and
interconnects

VHDL Design 
Styles

DATAFLOW

“concurrent”
statements • Sequential Logic 

(registers,
counters, etc.)

• State machines 
•Complex Comb. 
Logic

“sequential” statements

NON-
SYNTHESIZABLE

SYNTHESIZABLE

BEHAVIORAL

• Test Benches
• Modeling IP• Gates

• Simple Comb. 
Logic

architectureARCHITECTURE_NAME of ENTITY_NAME is

• Here you can declare signals, constants, functions, procedures…
• Component declarations
• No variable declarations !!

begin
Concurrent statements:

• Concurrent simple signal assignment 
• Conditional signal assignment 
• Selected signal assignment
• Generate statement

• Component instantiation statement

• Process statement
• inside process you can use only sequential                      

statements

end ARCHITECTURE_NAME;

Mixed Style Modeling

Concurrent Statements

(i.e. two processes are
concurrent)


